

Introduction à Maple : microéconomie du consommateur

Bernard Dupont et Frédéric Dupont

Bernard.Dupont@univ-lille1.fr

Ce chapitre donne un exemple des possibilités de Maple dans le domaine de la microéconomie. Comme dans le chapitre précédent, l'exemple est volontairement simple puisqu'il s'agit de la théorie statique du consommateur. A travers les résultats renvoyés par le logiciel, on peut donc se convaincre aisément de son intérêt dans les domaines suivants :

- 1- représentation graphique des fonctions de 2 variables réelles,
- 2- calcul de dérivées partielles, de dérivées logarithmiques et d'élasticités,
- 3- optimisation statique.

▼ Définition et représentation de la fonction d'utilité

L'avantage de Maple est qu'il est possible de visualiser très facilement et rapidement une fonction de deux variables réelles.


On commence par définir une Cobb-Douglas en toute généralité (les exposants sont α et β).

```
> util:=(x,y,alpha,beta)->x^(alpha)*y^(beta);#définition de la  
fonction d'utilité U(x,y,alpha,beta) qu'on appelle "util".
```

$$util := (x, y, \alpha, \beta) \rightarrow x^\alpha y^\beta$$


On demande le tracé de la surface représentant la valeur de l'utilité en fonction des quantités possédées de x et de y , d'abord pour $\alpha = \frac{1}{2}$ et $\beta = \frac{1}{2}$:

```
> plot3d(util(x,y,1/2,1/2),x=0..10,y=0..10,axes=boxed,  
numpoints=1000,orientation=[-66,58]);
```




ensuite pour $\alpha = \frac{1}{3}$ et $\beta = \frac{2}{3}$:

```
> plot3d(util(x,y,1/3,2/3),x=0..10,y=0..10,axes=boxed,
numpoints=1000,orientation=[-66,58]);
```




Quel est le niveau d'utilité lorsque l'individu ne possède pas de x ?

```
> plot(util(0,y,1/2,1/2),y=0..10,color=red,thickness=2);
```




Cela vous semble-t-il normal que, à cause de l'absence totale de x dans le panier de biens, le niveau d'utilité reste scotché à zéro alors que l'individu possède des quantités croissantes de y ?
Que pensez-vous de la variante suivante ?

```
> util1:=(x,y,alpha,beta)->((x+1)^(alpha)*(y+1)^(beta))-1;
#définition d'une autre fonction d'utilité, appelée "util1"
      util := (x, y, α, β) → (x + 1)α (y + 1)β - 1
> plot3d(util1(x,y,1/2,1/2),x=0..10,y=0..10,axes=boxed,
numpoints=1000,orientation=[-66,58]);
```



Cette nouvelle fonction d'utilité évacue le défaut de la précédente. Cette fois, l'utilité croît avec les quantités consommées de y même si le consommateur reste totalement privé de bien x comme le confirme le graphe suivant :

```
> plot(util1(0,y,1/2,1/2),y=0..10,color=red,thickness=2);
```



Il y a une morale à tirer de ce petit exercice. Maple est si facile d'emploi et si rapide qu'il est très aisé d'expérimenter des variantes d'une hypothèse de travail.

▼ Courbes d'indifférence et taux marginal de substitution

Le taux marginal de substitution est obtenu en appliquant le théorème de dérivation des fonctions implicites. Maple possède une procédure de différenciation implicite. On commence par poser une équation stipulant que le niveau d'utilité est constant, ce qui définit implicitement y en fonction de x (étant entendu que la fonction d'utilité a les propriétés requises).

```
> F:=U-util(x,y,alpha,beta)=0;
```


$$F := U - x^\alpha y^\beta = 0 \quad (2.1)$$

L'équation F est satisfaite par l'ensemble des couples (x,y) solutions de $U - x^\alpha y^\beta = 0$. Lorsqu'on fixe le niveau d'utilité, par exemple $U = 1$, ces couples définissent *une* courbe d'indifférence, c'est à dire une courbe regroupant tous les paniers de biens procurant à l'individu un même niveau d'utilité $U = 1$. On peut facilement tracer de telles courbes d'indifférence en utilisant la procédure

`implicitplot` de la bibliothèque (ou "paquetage", traduction officielle de l'anglais package) `plots`.

```
> with(plots):#chargement du paquetage plots
```

```
> implicitplot(subs(alpha=1/2,beta=1/2,U=1,F),x=0..2,y=0..2);
#courbe d'indifférence pour U=1 et alpha=beta=1/2.
```



On passe alors au calcul du taux marginal de substitution. On rappelle que le taux marginal de substitution *pour un panier donné de biens* indique la quantité supplémentaire de y ($dy > 0$) qui compense la perte d'une "petite" quantité de x ($dx < 0$) et ce de façon à maintenir constant le niveau d'utilité. Plus précisément, il s'agit de la limite de la valeur absolue du rapport de ces deux quantités lorsque dx tend vers zéro. On applique la procédure **implicitdiff(équation,variable 1,variable2)** pour obtenir la forme générique du taux marginal de substitution.

```
> tms:=abs(implicitdiff(F,y,x));
```

$$tms := \left| \frac{\alpha y}{\beta x} \right|$$

On considère généralement que le taux marginal de substitution est décroissant le long d'une même courbe d'indifférence. On peut le montrer de la façon suivante.


Établissons l'équation explicite d'une courbe d'indifférence (c'est à dire exprimons y en fonction de x) :

```
> indiff:=solve(F,y);
```

$$indiff := e^{-\frac{\ln(U) + \alpha \ln(x)}{\beta}}$$

On trace ensuite l'évolution du taux marginal de substitution sur cette courbe d'indifférence.

```
> plot(subs(y=indiff,alpha=1/2,beta=1/2,U=1,tms),x=0.1..3,
color=blue,numpoints=1000);
```



Et on vérifie que, pour les valeurs choisies, il est bien décroissant. On peut interpréter ce résultat de la façon suivante : lorsque la quantité de x est faible, le tms est important. cela veut dire que l'individu ne cédera du bien x (qu'il possède en faible quantité) que s'il reçoit en contrepartie beaucoup de y (suffisamment pour le dédommager de la perte du peu de x qu'il possède). En revanche, quand il dispose de beaucoup de x , il est prêt à en céder ... même contre une quantité négligeable de y . Cette interprétation est claire si on se souvient que le tms n'est rien d'autre que le rapport de l'utilité marginale de x et de y .

▼ Fonctions de demande

Passons à l'étude des fonctions de demande, des fonctions de demande inverse et des élasticités.


▼ Fonctions de demande

On commence par définir l'ensemble budgétaire d'un consommateur. L'ensemble budgétaire d'un consommateur est l'ensemble des couples (x,y) qu'il peut acheter sur le marché aux prix en vigueur avec son revenu. Les quantités de biens sont évidemment positives (c'est à dire positives ou nulles). Par ailleurs, un consommateur ne peut dépenser plus que son revenu R qui est positif (il peut dépenser (i) moins que ou (ii) exactement son revenu).

Avec Maple, il est facile de tracer l'ensemble budgétaire, qui est la surface en rouge dans le graphe suivant :

```
> inequal( {0<=x,0<=y,3*x+2*y<=20},x=-5..15,y=-5..15,
optionsfeasible=(color=red),
optionsclosed=(color=yellow,thickness=2),
optionsexcluded=(color=cyan),title="Ensemble budgétaire du
consommateur",labels=["x","y"]);
```

Ensemble budgétaire du consommateur



Le consommateur rationnel cherchera parmi l'ensemble des paniers accessibles celui qui lui procure le maximum d'utilité. Il cherche donc dans la zone rouge le couple (x,y) qui lui procure le maximum d'utilité. Pour trouver ce maximum, on programme une procédure pour résoudre ce problème avec la méthode de Lagrange :

```
> x:='x':#réinitialisation de la variable x
y:='y':#réinitialisation de la variable y
lagrange:=util(x,y,alpha,beta)+lambda*(R-px*x-py*y);
#définition du lagrangien
```

$$lagrange := x^\alpha y^\beta + \lambda (R - px x - py y)$$

Pour repérer les points candidats et leur multiplicateur associé, il faut annuler le gradient du lagrangien. Après avoir défini une routine de calcul de dérivée partielle, on pose les trois conditions du premier ordre.

```
> diffe:=(g,x)->diff(g,x);#définition d'une routine de calcul
de dérivée partielle d'une variable quelconque sur une
fonction g quelconque
```

$$diffe := (g, x) \rightarrow \frac{\partial}{\partial x} g$$

```
> eq:=map2(diffe=0,lagrange,{x,y,lambda});#en mapant2 la
```

relation "diffe"=0 sur la fonction "lagrange" et sur les trois variables x , y et λ on obtient les trois conditions du premier ordre sous forme d'ensemble

$$eq := \left\{ \frac{x^\alpha \alpha y^\beta}{x} - \lambda px = 0, \frac{x^\alpha y^\beta \beta}{y} - \lambda py = 0, R - px x - py y = 0 \right\}$$

Il ne reste plus qu'à résoudre ce système en ne conservant que les solutions en x et y positives.

```
> assume(x,nonnegative);assume(y,nonnegative):
sol:=solve(eq,{x,y,lambda});
```

$$sol := \left\{ \begin{aligned} x &= \frac{R \alpha}{(\alpha + \beta) px}, y = \frac{R \beta}{py (\alpha + \beta)}, \lambda \\ &= \frac{e^{\ln\left(\frac{R \beta}{py (\alpha + \beta)}\right) \beta + \ln\left(\frac{R \alpha}{(\alpha + \beta) px}\right) \alpha}}{R (\alpha + \beta)} \end{aligned} \right\}$$

La récupération des solutions en x et y permet de créer les fonctions de demande en x et y .

```
> xsol:=subs(sol,x);ysol:=subs(sol,y);#on affecte à xsol et
ysol les solutions du système précédent
```

$$xsol := \frac{R \alpha}{(\alpha + \beta) px}$$

$$ysol := \frac{R \beta}{py (\alpha + \beta)}$$

```
> demx:=unapply(xsol,px,R,alpha,beta);#fonction de demande de
x.
```


$$demx := (px, R, \alpha, \beta) \rightarrow \frac{R \alpha}{(\alpha + \beta) px}$$

```
> demy:=unapply(ysol,py,R,alpha,beta);#fonction de demande de
y.
```

$$demy := (py, R, \alpha, \beta) \rightarrow \frac{R \beta}{py (\alpha + \beta)}$$


A partir de ces fonctions, bien des choses sont possibles. On peut commencer par tracer la fonction de demande de x pour des valeurs données des paramètres α , β et R .

```
> plot(demx(px,10,1/2,1/2),px=0.01..2,x=0..20);#tracé de la
fonction de demande pour des valeurs fixées des paramètres
```

Et de même pour la demande de y :

> plot(demy(py,10,1/2,1/2),py=0.01..2,y=0..20);#tracé de la fonction de demande pour des valeurs fixées des paramètres



Ou encore : que se passe t-il quand le revenu du consommateur est modifié alors que les paramètres restent constants?


Calculons l'impact d'une augmentation de revenu sur la demande du bien x :

> diff(demx(px,R,alpha,beta),R);

$$\frac{\alpha}{(\alpha + \beta) px}$$

Cette dérivée partielle est positive dès l'instant où px , α , β le sont. Bref, une augmentation du revenu se traduit par une augmentation des quantités demandées. L'animation ci-dessous montre que lorsque le revenu augmente, la totalité de la courbe de demande se déplace vers la droite. Cela traduit le fait que, quel que soit le prix, les quantités demandées sont plus importantes lorsque R augmente.

```
> with(plots):#chargement du paquetage plots
animate( plot,[demx(px,R,1/2,1/2),px=0.01..0.2,x=500.
.2000],R=10..150);#création d'une animation
```



Pour lancer l'animation, il faut "cliquer gauche" sur le graphique, puis cliquer sur le triangle jaune ombré dans le menu contextuel.

▼ Fonctions de demande inverse

On peut calculer et tracer une demande inverse (c'est à dire les prix en fonction des quantités). C'est la façon habituelle de présenter une courbe de demande dans le monde anglo-saxon.

Dans ce qui suit, on s'en tient à la demande de x .

```
> sol1:=solve(demx(px,R,alpha,beta)=x,{px});#inversion de la
fonction de demande de x.
```

$$sol1 := \left\{ px = \frac{R \alpha}{x \sim (\alpha + \beta)} \right\}$$

On voit qu'il y a une hypothèse sur x à cause du tilde. Quelle est cette hypothèse?

```
> about(x);
```

Originally x , renamed $x\sim$:
is assumed to be: `RealRange(0,infinity)`

Maple n'a pas oublié que les quantités achetées et consommées doivent être positives ou nulles.
On peut récupérer sans crainte la solution,

```
> pxsol:=subs(sol1,px);
```

$$pxsol := \frac{R \alpha}{x\sim (\alpha + \beta)}$$


définir la fonction de demande inverse du bien x ,

```
> demx1:=unapply(pxsol,x,R,alpha,beta);#on crée la fonction  
de demande qui dépend des prix, du revenu et des paramètres  
alpha et beta
```

$$demx1 := (x\sim, R, \alpha, \beta) \rightarrow \frac{R \alpha}{(\alpha + \beta) x\sim}$$

et en donner une représentation graphique.

```
> plot(demx1(x,10,1/2,1/2),x=0..4,px=0..12);
```



▼ Elasticités

Connaissant les fonctions de demande de x et de y , on peut calculer leurs élasticités.

Demandons l'élasticité de la demande de x par rapport à son prix :

```
> elaspd:=diff(demx(px,R,alpha,beta),px)*(px/demx(px,R,alpha,  
beta));
```

$$elpsd := -1$$

Puis l'élasticité de la demande de x par rapport au prix de y (élasticité-prix croisée) :

```
> elaspc:=diff(demx(px,R,alpha,beta),py)*(py/demx(px,R,alpha,
beta));
                                     elaspc:=0
```

Enfin, l'élasticité de la demande par rapport au revenu :

```
> elasR:=diff(demx(px,R,alpha,beta),R)*(R/demx(px,R,alpha,
beta));
                                     elasR:=1
```

On peut d'ailleurs automatiser le calcul en utilisant la fonction **map2**. On voit qu'on obtient facilement des résultats issus de calculs répétitifs.

Elasticités par rapport au prix du bien, du prix de l'autre bien et du revenu de la demande de x :

```
> elasx:=map2((f,x)->diff(f,x)*(x/f),demx(px,R,alpha,beta),
[px,py,R]);
                                     elasx:= [-1, 0, 1]
```

Idem pour les élasticités de la demande de y :

```
> elasy:=map2((f,x)->diff(f,x)*(x/f),demy(px,R,alpha,beta),
[px,py,R]);
                                     elasy:= [-1, 0, 1]
```

▼ Conclusion

Encore plus que celui du chapitre précédent, ce worksheet doit convaincre le lecteur de l'intérêt d'utiliser Maple comme auxiliaire actif de la modélisation économique. D'une part, les difficultés techniques sont plus importantes : les fonctions utilisées en microéconomie sont rarement linéaires et il y est moins question d'algèbre que d'analyse, de préférence dans le domaine des fonctions de plusieurs variables réelles; les représentations graphiques sont en deux ou trois dimensions; l'optimisation est omniprésente et recourt massivement à la méthode de Lagrange. Le logiciel n'a séché sur aucun des problèmes posés : calculs de dérivées partielles et d'élasticités; résolution de systèmes non linéaires; graphiques en 3D. D'autre part, on a engrangé une nouvelle leçon de méthode expérimentale. Un gros avantage du logiciel est qu'il permet de tester très facilement des hypothèses de travail, en particulier des hypothèses alternatives à celles communément admises. On dispose ainsi d'un collaborateur docile et puissant pour la recherche, à condition de bien maîtriser son potentiel. Les économistes ont donc tout intérêt à dominer la programmation.